# Maintainer: Bartłomiej Piotrowski <bpiotrowski@archlinux.org>

pkgname=rpcsvc-proto
pkgver=1.4.2
pkgrel=1
pkgdesc='rpcsvc protocol definitions from glibc'
arch=(x86_64)
url='https://github.com/thkukuk/rpcsvc-proto'
license=(BSD)
depends=(glibc)
source=($pkgname-$pkgver.tar.gz::https://github.com/thkukuk/rpcsvc-proto/archive/v$pkgver.tar.gz)
sha512sums=('6769f9439e3f187eebdeef4ee8d54f8a6fee6f410e3137d0c1b26e61b705873932890856faff55b68c39aa702e456b36fe9410b85baf1ef9b20ee97f2158971a')

prepare() {
  cd $pkgname-$pkgver
  autoreconf -fi
}

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
  install -Dm644 COPYING "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
